

ساختار جملات و توابع شرطی

جملات شرطی که در گفتگوهای روزمره مورد استفاده قرار می گیرند، دارای ساختار زیر می باشد:

قسمت اول (جمله شرطی) شامل:

کلمه شرط - **خود شرط** - **نتیجه شرط**

قسمت دوم (جمله نقیض) شامل:

نقیض شرط - **نتیجه نقیض شرط** - **پایان شرط**

در محاورات روزمره، معمولاً قسمت دوم را بر اساس قرینه معنوی حذف می کنیم زیرا شنونده می تواند نقیض گزاره شرطی را حدس بزند. اما در بعضی موارد قسمت دوم جمله شرطی را هم استفاده می نمائیم که این کار دو دلیل دارد:

1 - ارزش قسمت دوم (جمله نقیض) بیش از جمله اول است.

2 - قصد تعیین تکلیف برای شنونده را داریم.

به جمله زیر دقت کنید:

" اگر هوا تاریک است، چراغ را روشن کن "

این جمله، یک جمله شرطی است که فاقد قسمت **نقیض** می باشد. اگر شنونده چراغ را روشن کند، بدلیل آن است که شرط اتفاق افتاده است و اگر آن را روشن نکند، نتیجه گرفته است که در صورت **تاریک نبودن هوا**، نیازی به روشن کردن چراغ نیست.

تاریک نبودن هوا را بعنوان نقیض شرط، و **روشن نکردن** چراغ را بعنوان نتیجه نقیض معرفی می نمائیم.

حال به جمله زیر دقت کنید:

" اگر والدین برای بردن تو به مدرسه آمدند که هیچ، در غیر اینصورت مبادا از مدرسه خارج شوی! "

این جمله، یک جمله شرطی است که دارای قسمت **نقیض** هم می باشد. اما پر واضح است که اهمیت جمله دوم (جمله نقیض) بسیار بیشتر از جمله اول (جمله شرط) بوده است که آن را بصورت تأکیدی آورده ایم.

ساختار توابع شرطی در نرم افزار اکسل، دقیقاً مشابه ساختاری است که در محاورات روزمره استفاده می کنیم، با این تفاوت که چون کامپیوتر فاقد قدرت تحلیل و نتیجه گیری هوشمندانه

می باشد، در صورت عدم مشخص نمودن نقیض شرط دچار بلاتکلیفی میگردد. بنابراین به هنگام استفاده از توابع شرطی ترجیحاً نقیض شرط را هم مشخص می نمائیم تا در صورتی که تابع شرطی به نقیض شرط رسید، دچار سردرگمی نشود. جهت تطابق اجزاء جملات شرطی در زبان های فارسی و انگلیسی، و همچنین معادل آنها در توابع اکسل به جدول زیر توجه کنید:

Excel	جمله انگلیسی	شرح	جمله فارسی	شرح
If (If	---	اگر	کلمه شرط
.....	The Weather is Cold,	اینگونه (بود، هست، شد)	هوا سرد است،	خود شرط
;	Then Turn On The Heater,	آنگاه اینکار را بکن	بخاری را روشن کن،	نتیجه شرط
;	Else,	وگرنه	در غیر اینصورت،	نقیض شرط
.....	Don't Turn On The Heater	این کار را بکن	بخاری را روشن نکن	نتیجه نقیض شرط
)	.	پایان	.	پایان شرط

گزینه هائی که با رنگ سفید نوشته شده اند را می توان حذف نمود بدون اینکه به ساختار جمله لطمه ای وارد شود. فرض کنید بخواهیم جمله زیر را به زبان توابع اکسل بنویسیم:

اگر مقدار موجود در خانه C5 بزرگتر از دو برابر مقدار موجود در خانه A2 بود، آنگاه نصف مقدار موجود در خانه B2 را محاسبه کرده و بنویس، در غیر این صورت، بنویس **Nothing!**.

با توجه به آنچه در بالا گفته شد این جمله را به صورت زیر می نویسیم:

=If (C5 > 2*A2 ; B2/2 ; "Nothing!")

تفکیک تابع فوق به اجزاء یک جمله شرطی به صورت زیر می باشد:

کلمه شرط	If (
خود شرط	بزرگتر بودن محتویات سلول C5 از دو برابر محتویات سلول A2
نتیجه شرط	محاسبه و درج نصف محتویات سلول B2 ;
نقیض شرط	;
نتیجه نقیض شرط	درج کلمه Nothing
پایان شرط)

توابع شرطی یکی از مهمترین گروه توابع اکسل می باشند که آشنائی و به کار بستن آنها باعث افزایش توانمندی کاربران و بکارگیری قابلیت های وسیع این نرم افزار میگردد. به همین دلیل توصیه میگردد این گروه از توابع را با اهمیت بیشتری دنبال نمائید.

1- تابع If

شرطی را در **سلول دیگری** کنترل نموده و در صورت **إرضاء آن**، **نتیجه شرط** را باز میگردداند. در غیر اینصورت **ضد شرط** را اجرا می نماید.

(ضد شرط ; نتیجه شرط ; موضوع شرط) =If : فرم تابع

	A	B	C	D
1				
2		5		
3		6		
4		2		
6		نام تابع	خواسته	فرمول
7		IF	اگر مقدار خانه بنفش رنگ کوچکتر از 5 بود خودش را بنویس در غیر اینصورت دو برابر آن را محاسبه کن.	=IF(B2<5;B2;2*B2)

	A	B	C	D
1				
2		5		
3		6		
4		2		
6		نام تابع	خواسته	
9		IF	اگر در خانه سبز رنگ کلمه Ali مشاهده کردی آن را بنویس در غیر اینصورت بنویس: مشترک مورد نظر در دسترس نمی باشد!	=IF(B3="Ali";B3;"!مشترک مورد نظر در دسترس نمی باشد!")

نکته - 1: توجه داشته باشید که در توابع شرطی **موضوع شرط** در **سلول دیگری** می باشد نه در همان سلول که فرمول را می نویسیم.

	A	B	C
1			
2		5	
3		6	
4		2	
6		نام تابع	خواسته
11		IF	اگر در خانه آبی رنگ عددی کمتر یا مساوی ۱۰ مشاهده کردی آن عدد را در ۱۰ ضرب کن، اگر عددی بین ۱۰ و ۲۰ مشاهده کردی آن را در ۲۰ ضرب کن، اگر خود ۲۰ را مشاهده کردی آن را نصف کن، در غیر اینصورت ۳ برابر آن را با عدد ۹۰ مقایسه کن و بیشترین عدد را درج کن!
12			فرمول
13			=IF(B4<=10;B4*10;IF(B4<20;B4*20;IF(B4=20;B4/2;MAX(B4*3;90))))

	A	B	C
1			
2		5	
3		6	
4		2	
6		نام تابع	خواسته
15		IF	اگر مجموع اعداد خانه های بنفش و سبز و آبی کوچکتر یا مساوی ۵۰ بود کوچکترین آنها، اگر کوچکتر یا مساوی ۱۰۰ بود متوسط آنها، و اگر بزرگتر از ۱۰۰ بود بزرگترین آنها را درج کن!
16			فرمول
17			=IF(SUM(B2;B3;B4)<=50;MIN(B2;B3;B4);IF(SUM(B2;B3;B4)<=100;AVERAGE(B2;B3;B4);MAX(B2;B3;B4)))

	A	B	C
1			
2		5	
3		6	
4		2	
6		نام تابع	خواسته
19		IF	کوچکترین عدد موجود در خانه های بنفش و سبز و آبی را به همراه رنگ آن گزارش کن!
20			فرمول
21			=IF(MIN(B2;B3;B4)=B2;"Violet = "&B2;IF(MIN(B2;B3;B4)=B3;"Green = "&B3;"Blue = "&B4))

کلیه مثال های بالا به صورت کامل نوشته شده اند. یعنی دارای نقیض شرط نیز می باشند. همانگونه که قبلاً نیز توضیح داده شد، بهتر است توابع شرطی را بصورت کامل بنویسیم تا در صورت مواجهه تابع با نقیض شرط، برای تابع تعیین تکلیف شده باشد.

بهترین شیوه برای درک بهتر توابع شرطی تمرین می باشد. بنابراین سعی کنید تا میتوانید تابع شرطی بنویسید.

نکته - 2: توجه داشته باشید که آرگومان های هر تابع می تواند یک تابع دیگر باشد. بر همین اساس می توان توابع **شرطی تو در تو** نوشت. حداکثر تعداد شرط هائی که می توان بصورت تو در تو نوشت **7 شرط** می باشد.

نکته - 3: هر تابع شرطی می تواند **تنها یک شرط** را کنترل کند. بنابراین چنانچه صورت مسئله دارای بیش از یک شرط باشد، بایستی از **شرط های تو در تو** و یا سایر توابع شرطی (And ; Or) استفاده نمود.

-2 تابع And

چند شرط را در **سلول(ها) دیگر(ی)** کنترل می نماید و در صورت **إرضاء همه** آنها، **True** و در غیر اینصورت **False** را باز می گرداند.

(..... ; شرط دوم ; شرط اول) And =: فرم تابع

-3 تابع Or

چند شرط را در **سلول(ها) دیگر(ی)** کنترل می نماید و در صورت **إرضاء تنها یکی از آنها**، **True** و در غیر اینصورت **False** را باز می گرداند.

(..... ; شرط دوم ; شرط اول) Or =: فرم تابع

نکته - 4: توابع And و Or به تنهایی دستاورد قابل ملاحظه ای ندارند و صرفاً یک اطلاع رسانی را در پی خواهند داشت. اما اگر نتیجه این توابع را بعنوان یکی از آرگومان های تابع If بکار ببریم، قابلیت های تابع If بطور قابل ملاحظه ای افزایش یافته و در بسیاری از موارد از نوشتن شرط های تو در تو که علاوه بر محدودیت تعداد (حداکثر 7 شرط) از پیچیدگی بیشتری هم برخوردار می باشند، بی نیاز می شویم.

	A	B	C
1			
2		5	
3		6	
4		2	
6		نام تابع	خواسته
24		IF / AND	اگر مجموع اعداد خانه های بنفش و سبز و آبی کوچکتر یا مساوی ۵۰ بود و هر سه آنها مثبت بودند، بزرگترین آنها را همراه با کلمه "Max =" درج کن. در غیر این صورت بنویس: "گشتم نبود نگرد که نیست!"
25			فرمول
26			=IF(AND(SUM(B2;B3;B4)<=50;B2>0;B3>0;B4>0);"Max="&MAX(B2;B3;B4);"گشتم نبود نگرد که نیست!")

	A	B	C
1			
2		5	
3		6	
4		2	
6		نام تابع	خواسته
29		IF / OR	اگر در خانه سبز فقط کلمه Ali را چه به زبان فارسی و چه به زبان انگلیسی یافتی آن را نمایش بده.
30			فرمول
31			=IF(OR(B3="Ali";B3="علی");B3;"")

	A	B	C
1			
2		5	
3		6	
4		2	
6		نام تابع	خواسته
33		IF / OR	اگر هر یک از اعداد موجود در خانه های بنفش و سبز و آبی از مجموع اعداد دو خانه دیگر بزرگتر بود، همان عدد را درج کن ، در غیر این صورت مجموع آنها را بنویس.
34			فرمول
35			=IF(OR(B2>SUM(B3;B4);B3>SUM(B2;B4);B4>SUM(B2;B3));MAX(B2;B3;B4);SUM(B2;B3;B4))

4- تابع Sumif

شرطی را در ستون(سطری) از داده‌ها کنترل، در صورت **إرضاء** آن، مقادیر منطبق با شرط را در **همان ستون (سطر) یا ستون (سطر) دیگری** با هم جمع می‌کند.

(محدوده اعمال ; شرط مورد نظر ; محدوده جستجو) Sumif = فرم تابع

5- تابع Sumifs

چند شرط را در ستون های (سطرهای) متعددی از داده‌ها کنترل، در صورت **إرضاء همه آنها**، مقادیر منطبق با شرطها را در **همان ستون (سطر) یا ستون (سطر) دیگری** با هم جمع می‌کند.

(محدوده جستجو ; شرط های مورد نظر ; محدوده اعمال) Sumifs = فرم تابع

6- تابع Averageif

شرطی را در ستون(سطری) از داده‌ها کنترل، در صورت **إرضاء** آن، میانگین مقادیر منطبق با شرط را در **همان ستون (سطر) یا ستون (سطر) دیگری** باز میگرداند.

(محدوده اعمال ; شرط مورد نظر ; محدوده جستجو) Averageif = فرم تابع

7- تابع Averageifs

چند شرط را در ستون های (سطرهای) متعددی از داده‌ها کنترل، در صورت **إرضاء همه آنها**، میانگین مقادیر منطبق با شرطها را در **همان ستون (سطر) یا ستون (سطر) دیگری** باز میگرداند.

(محدوده جستجو ; شرط های مورد نظر ; محدوده اعمال) Averageifs = فرم تابع

8- تابع Countif

شرطی را در ستون (سطری) از داده‌ها کنترل، در صورت **إرضاء** آن، تعداد اقلام منطبق با شرط را در **همان ستون (سطر)** شمارش می‌کند.

(شرط مورد نظر ; محدوده جستجو) **Countif =** فرم تابع

9- تابع Countifs

چند شرط را در ستون های (سطرهای) متعددی از داده‌ها کنترل، در صورت **إرضاء همه آنها**، تعداد اقلام منطبق با شرطها را در **همان ستون (سطر)** شمارش می‌نماید.

(شرط های مورد نظر ; محدوده جستجو) **Countifs =** فرم تابع

ردیف	نام	ماه	روز	مبلغ (ریال)
1	A	2	23	800000
2	B	3	17	300000
3	C	8	7	900000
4	D	5	25	900000
5	E	1	13	200000
6	F	12	20	700000
7	A	2	18	200000
8	B	6	5	700000
9	C	10	28	200000
10	C	1	2	600000
11	D	9	24	100000
12	F	6	10	900000
13	F	6	6	800000
14	E	6	29	500000
15	A	8	24	300000
16	B	6	3	500000
17	A	12	15	500000

نام تابع	خواسته	فرمول	نتیجه
Sumif	مجموع پرداخت های "A"	=SUMIF(C3:C19;"A";F3:F19)	1800000
Countif	تعداد پرداخت های "F"	=COUNTIF(C3:C19;"F")	3
Averageif	متوسط پرداخت های "C"	=AVERAGEIF(C3:C19;"C";F3:F19)	566667
Sumifs	مجموع پرداخت های "A" در شش ماهه اول سال	=SUMIFS(F3:F19;C3:C19;"A";D3:D19;"<=6")	1000000
Countifs	تعداد پرداخت های "F" در دهه اول نیمه دوم سال	=COUNTIFS(C3:C19;"F";E3:E19;"<=10";D3:D19;">6")	0
Averageifs	متوسط پرداخت های "D" در نیمه اول سال	=AVERAGEIFS(F3:F19;C3:C19;"D";D3:D19;"<7")	900000

نکته - 5: هر گاه معرفی یکی از شروط توابع شمارش شرطی نیازمند استفاده از علائم < یا > باشد، بایستی حتماً آن شرط را درون گیومه قرار داد.